
Nagyhatékonyságúlogikai programozás

A tárgy témakörei

� Korlát-logikaiprogramozás(CLP — ConstraintLogic Programming)
� A Mercury„nagybani”logikai programozásinyelv

Inf ormációk a korlát-logikai programozásról

� „Sárgakönyv”: Kim Mariott, PeterJ.Stuckey, Programmingwith
Constraints:anIntroduction,MIT Press1998(részletesebbenlásd
http://www.cs.mu.oz.au/ ~pjs /bo ok/b ook. html )
� „Az els̋o alapkönyv”: PascalVanHentenryck:ConstraintSatisfactionin

Logic Programming,MIT Press,1989
� On-lineGuideto ConstraintProgramming,by RomanBarták

(http://kti.ms.mff.cuni .cz/ ~bar tak /con stra ints / )
� Korlát-programozásiarchívum:

http://www.cs.unh.edu/c cc/a rch ive

Inf ormációk a Mercury nyelvről

� Honlap:http://www.cs.mu.oz.au/r esea rch /mer cury /

A CLP alapgondolata

A CLP(
�

) séma

Prolog+
egy valamilyen

�
adattartományraésazonértelmezettkorlátokra(re-

lációkra)vonatkozó„erős” következtetésimechanizmus.

Példákaz
�

tartomány megválasztására

� �
Q vagyR (a racionálisvagyvalósszámok)

korlátok
�

lineárisegyenl̋oségekésegyenl̋otlenségek
következtetésimechanizmus

�
Gaußeliminációésszimplex módszer

� �
FD (egészszámokVégesTartománya,angolulFD — FiniteDomain)

korlátok
�

különfélearitmetikaiéskombinatorikusrelációk
következtetésimechanizmus

�
MI CSP–módszerek(CSP= Korlát-Kielégítési

Probléma)

� �
B (0 és1 Booleértékek)

korlátok
�

ítéletkalkulusbelirelációk
következtetésimechanizmus

�
MI SAT-módszerek(SAT — Boole

kielégíthet̋oség)

OR ADAI... ...

ResolutionCSPSimplex ... ...

CLP(H)
CLP(R)

Logic
Programming

Prolog =
CLP(FD)

Mercury = CLP(0)

Példa: CLP(MiniNat)

Egy miniat űr kvázi-CLP nyelv természetesszámokra

� Tartomány: Nemnegatív egészek
� Függvények:

+ - *

� Korlát-relációk:

= < > =< >=

� Korlát-megoldóalgoritmus:

SICStuskorutin-kiterjesztésénalapul

A Prologbaágyazásszintaxisa:

{ Korlát} aKorlát felvétele
({X} szintaktikusédesít̋oszer, ekvivalensa ’{}’(X) kifejezéssel.)

Példafutás

| ?- {2*X+3*Y=8}.
X = 1, Y = 2 ? ;
X = 4, Y = 0 ? ;
no
| ?- {X*2+1=28}.
no
| ?- {X*X+Y*Y=25, X > Y}.
X = 4, Y = 3 ? ;
X = 5, Y = 0 ? ;
no

Prolog háttér: blokkolás,korutinszervezés

Blokk-deklarációk SICStusban
Egyeljárásraelőírhatjuk,hogymindaddig,amígegy ún.blokkolásifeltétel
fennáll,azeljárásfüggeszt̋odjékfel. Példa:

:- block p(-, ?, -, ?, ?).

Jelentése:haazels̋o ésaharmadikargumentumis behelyettesítetlenváltozó
(blokkolásifeltétel),akkor ap hívásfelfüggeszt̋odik.

Ugyanarraazeljárásratöbbvagylagosfeltételis szerepelhet,pl.
:- block p(-, ?), p(?, -).

Blokk-deklarációk haszna

� Adatfolyam-programozás(lásdHammingprobléma,Prologjegyzet)
� Generálésellen̋oriz programokgyorsítása
� Végtelenválasztásipontokkiküszöbölése

Biztonságosappend/3 , blokk-deklarációval

:- block append(-, ?, -).
% blokkol, ha az els ő és a harmadik argumentum
% egyaránt behelyettesítetlen
append([], L, L).
append([X|L1], L2, [X|L3]) :-

append(L1, L2, L3).

5



Példakorutinszervezésre: többirányú összeadás

% X+Y=Z, ahol X, Y és Z természetes számok.
% Bármelyik argumentum lehet behelyettesítetlen.
plusz(X, Y, Z) :-

append(A, B, C),
len(A, X),
len(B, Y),
len(C, Z).

% L hossza Len.
len(L, Len) :-

len(L, 0, Len).

:- block len(-, ?, -).
% L lista hossza Len-Len0. Len0 mindig ismert.
len(L, Len0, Len) :-

nonvar(Len), !, Len1 is Len-Len0,
length(L, Len1).

len(L, Len0, Len) :-
% nonvar(L), % a blokkolási feltétel miatt!
( L == [] -> Len = Len0
; L = [_|L1],

Len1 is Len0+1, len(L1, Len1, Len)
).

| ?- plusz(X, Y, 2).
X = 0, Y = 2 ? ;
X = 1, Y = 1 ? ;
X = 2, Y = 0 ? ;
no
| ?- plusz(X, X, 8).
X = 4 ? ;
no
| ?- plusz(X, 1, Y), plusz(X, Y, 20).
no

További korutinszervez̋o eljárások

Hívásokkésleltetése

freeze(X, Hivas)
Hivas t felfüggesztimindaddig,amígX behelyettesítetlenváltozó.

dif(X, Y)
X ésY nemegyesíthet̋o. Mindaddigfelfüggeszt̋odik, amígezel nemdönthet̋o.

when(Feltétel, Hívás)
Blokkolja aHívás t mindaddig,amígaFeltétel igazzánemválik. Itt a
Feltétel egy (nagyon)leegyszer̋usítettPrologcél,amelynekszintaxisa:

CONDITION ::= nonvar(X) | ground(X) | ?=(X,Y) |
CONDITION, CONDITION |
CONDITION; CONDITION

(ground(X) jelentése:X, tömör, azaznemtartalmaz(behelyettesítetlen)
változót
?=(X,Y) jelentése:X ésY egyesíthet̋oségeeldönthet̋o.)
Példa:

| ?- when( ((nonvar(X);?=(X,Y)),gr ound (T) ),
process(X,Y,T)).

Késleltetetthívásoklekérdezése

frozen(X, Hivas)
Az X változómiatt felfüggesztetthívás(oka)tegyesítiHivas -sal.

call_residue(Hivas, Maradék)
Hivas -t végrehajtja,éshaasikereslefutásutánmaradnakfelfüggesztett
hívások,akkor azokatvisszaadjaMaradékban . Pl.

| ?- call_residue((dif(X, f(Y)), X=f(Z)), Maradek).
X = f(Z),
Maradek = [[Y,Z]-(prolog:dif(f(Z),f( Y))) ] ?

CLP(MiniNat) megvalósítása

Számábrázolás
� A korábbiplusz/3 eljárásbanegy � elem̋u listával ábrázoltukaz �

számot(a listaelemekérdektelenek,általábanbehelyettesítetlenváltozók)
� Példa:a2 számábrázolása:[_,_] � .(_,.(_,[])) .
� Hagyjukel a feleslegesváltozókat,akkor a2 számábrázolása:.(.([])) .
� Itt a [] jelenti a0 számot,a .(X) struktúraazX számrákövetkez̋ojét (a

nála1-gyelnagyobbszámot).
� Ez tulajdonképpenaPeanoféleszámábrázolás,haa . /1 helyettazs/1

funktort,a [] helyetta0 konstansthasználjuk.
� A CLP(MiniNat)megvalósításábanaPeanoszámábrázolásthasználjuk,

tehát;0 = 0; 1 = s(0); 3 = s(s(s(0))) stb.

Összeadáséskivonás

% plusz(X, Y, Z): X+Y=Z (Peano számokkal).
:- block plusz(-, ?, -).
plusz(0, Y, Y).
plusz(s(X), Y, s(Z)) :-

plusz(X, Y, Z).

% +(X, Y, Z): X+Y=Z (Peano számokkal). Hatékonyabb, mert
% továbblép, ha bármelyik argumentum behelyettesített.
:- block +(-, -, -).
+(X, Y, Z) :-

var(X), !, plusz(Y, X, Z). % \+((var(Y),var(Z)))
+(X, Y, Z) :-

/* nonvar(X), */ plusz(X, Y, Z).

% X-Y=Z (Peano számokkal).
-(X, Y, Z) :-

+(Y, Z, X).

CLP(MiniNat) megvalósítása(folyt.)

A szorzásműveletmegvalósításielvei:
� Felfüggesztjükmindaddig,míg legalábbegy tényez̋o vagyaszorzatismertté

nemválik.
� Haazegyik tényez̋o ismert,visszavezetjükismételtösszeadásra.
� Haaszorzatismert( � ), azegyik tényez̋orevégigpróbáljukaz1,2,.. . �

értékeket,ezáltalismételtösszeadásravisszavezethet̋ovétesszük.

% X*Y=Z. Blokkol, ha nincs tömör argumentuma.
*(X, Y, Z) :-

when( (ground(X);ground(Y);groun d(Z )),
szorzat(X, Y, Z)).

% X*Y=Z, ahol legalább az egyik argumentum tömör.
szorzat(X, Y, Z) :-

( ground(X) -> szor(X, Y, Z)
; ground(Y) -> szor(Y, X, Z)
; /* Z tömör! */

Z == 0 -> szorzatuk_nulla(X, Y)
; +(X, _, Z), % X =< Z, vö. between(1, Z, X)

szor(X, Y, Z)
).

% X*Y=0.
szorzatuk_nulla(X, Y) :-

( X = 0 ; Y = 0 ).

% szor(X, Y, Z): X*Y=Z, X tömör.
% Y-nak az (ismert) X-szeres összeadása adja ki Z-t.
szor(0, _X, 0).
szor(s(X), Y, Z) :-

+(Z1, Y, Z),
szor(X, Y, Z1).

CLP(MiniNat) megvalósítása:(folyt. 2)

A korlátok végrehajtása
� A funkcionálisalakbanmegadottkorlátokata+ /3, - /3, * /3

hívásokbólálló célsorozattáalakítjuk,majdeztacélsorozatotmeghívjuk.
� Példáula {X*Y+2=Z} korlát lefordítottalakja:

*(X, Y, _A),+(_A, s(s(0)), Z) ,
� Az {X =< Y} korlátotaz{X+_ = Y} korlátra,az{X < Y} korlátot

pedigaz{X+s(_) = Y} korlátravezetjükvissza

% {Korlat}: Korlat fennáll.
{Korlat} :-

korlat_cel(Korlat, Cel), call(Cel).

Korlátok fordítása

% korlat_cel(Korlat, Cel): Korlat végrehajtható
% alakja a Cel célsorozat.
korlat_cel(Kif1=Kif2, (C1,C2)) :-

kiertekel(Kif1, E, C1),
kiertekel(Kif2, E, C2).

korlat_cel(Kif1 =< Kif2, Cel) :-
korlat_cel(Kif1+_ = Kif2, Cel).

korlat_cel(Kif1 < Kif2, Cel) :-
korlat_cel(s(Kif1) =< Kif2, Cel).

korlat_cel(Kif1 >= Kif2, Cel) :-
korlat_cel(Kif2 =< Kif1, Cel).

korlat_cel(Kif1 > Kif2, Cel) :-
korlat_cel(Kif2 < Kif1, Cel).

korlat_cel((K1,K2), (C1,C2)) :-
korlat_cel(K1, C1),
korlat_cel(K2, C2).

CLP(MiniNat) megvalósítása:(folyt. 3)

Kifejezésekfordítása
� EgyKif1 Op Kif2 kifejezéslefordítottalakjaegy háromrészb̋ol álló

célsorozat,amelyegy E változóbanállítja elő akifejezéseredményét:

– els̋o rész:Kif1 értékétpl. A-banelőállító cél(sororzat).

– másodikrész:Kif2 értékétpl. B-banelőállító cél(sororzat).

– harmadikrész:azOp(A, B, E) hívás(aholOpa+, -, * jelek
egyike).

� Egyszámlefordítottformájaaző Peanoalakja.
� Mindenegyéb(változó,vagymárPeanoalakúszám)változatlanmarada

fordításkor.

% kiertekel(Kif, E, Cel): A Kif aritmetikai kifejezés
% értékét E-ben el őállító cél Cel.
% Kif egészekb ől a +, -, és * operátorokkal épül fel.
kiertekel(Kif, E, (C1,C2,Rel)) :-

nonvar(Kif),
Kif =.. [Op,Kif1,Kif2], !,
kiertekel(Kif1, E1, C1),
kiertekel(Kif2, E2, C2),
Rel =.. [Op,E1,E2,E].

kiertekel(N, Kif, true) :-
number(N), !,
int_to_peano(N, Kif).

kiertekel(Kif, Kif, true).

% int_to_peano(N, P): N természetes szám Peano alakja P.
int_to_peano(0, 0).
int_to_peano(N, s(P)) :-

N > 0, N1 is N-1,
int_to_peano(N1, P).

Prolog háttér: kifejezésektestreszabottkiírása

print/1
Alapértelmezésbenazonoswrite -tal. Haa felhasználódefiniálegy
portray/1 eljárást,akkor a rendszermindenaprint -tel kinyomtatandó
részkifejezésremeghívjaportray -t. Enneksikereeseténfeltételezi,hogya
kiírásmegtörtént,meghiúsuláseseténmaga írja ki a részkifejezést.
A rendszeraprint eljárásthasználjaaváltozó-behelyettesítésekésa
nyomkövetéskiírására!

portray/1
Igaz,haKif kifejezéstaPrologrendszerneknemkell kiírnia. Alkalmas
formábankiírja aKif kifejezést.
Ezegy felhasználóáltal definiálandó(kampó) eljárás(hookpredicate).

% Peano számok kiírásának formázása
user:portray(Peano) :-

peano_to_int(Peano, 0, N), write(N).

% A Peano Peano-szám értéke N-N0.
peano_to_int(Peano, N0, N) :-

nonvar(Peano),
( Peano == 0 -> N = N0
; Peano = s(P),

N1 is N0+1,
peano_to_int(P, N1, N)

).

% felfüggesztett célok kiíratásának formázása
user:portray(user:Rel) :-

Rel =.. [Op,A,B,C],
( Op = (+) ; Op = (-) ; Op = (*) ),
Fun =.. [Op,A,B],
print({Fun=C}).

CLP(MiniNat) használata— példa

:- block fact(-,-).
fact(N, F) :-

{N = 0, F = 1}.
fact(N, F) :-

{N >= 1, N1 = N-1},
fact(N1, F1),
{F = N*F1}.

| ?- fact(6, F).
F = 720 ? ; no

| ?- fact(8, F).
F = 40320 ? ; no

| ?- fact(F, 6).
F = 3 ? ; no

| ?- fact(F, 24).
F = 4 ? ;
! Resource error: insufficient memory

| ?- fact(F, 10).
no

| ?- fact(F, 11).
! Resource error: insufficient memory

| ?- {X*X+Y*Y=25, X>Y}.
X = 4, Y = 3 ? ;
X = 5, Y = 0 ? ;
X = 5, Y = 0 ? ;
no

13



Prolog háttér: programok előfeldolgozása

Kampó eljárások a fordítási idejű átalakításhoz:

� term_expansion(+Kif, -Klózok) : Mindenbetölt̋o eljárás
(consult, compile stb.) által beolvasottkifejezésrea rendszer
meghívja.A második,kimenő paraméterbenvárjaa transzformáltalakot
(lehetlista is). Meghiúsuláseseténváltoztatásnélkül veszifel akifejezést
klózként.
� goal_expansion(+Cél, +Modul, -ÚjCél) : Mindenabeolvasott

programban(vagyfeltett kérdésben)előfordulórészcélrameghívjaa
rendszer. A harmadik,kimenő paraméterbenvárjaa transzformáltalakot
(lehetkonjunkció).Meghiúsuláseseténváltoztatásnélkül hagyjaacélt.

CLP(MiniNat) továbbfejlesztésegoal_expansion használatával

� A funkcionálisalakátalakításaabetöltésalatt is elvégezhet̋o:

user:goal_expansion({Ko rlat }, _, Cel) :-
korlat_cel(Korlat, Cel).

� A faktoriálispéldabetöltöttalakja(a true hívásokelhagyásaután):

fact(0, s(0)).
fact(N, F) :-

+(s(0), _, N), % N >= 1
-(N, s(0), N1), % N1 = N-1
*(N, F1, F), % F = N*F1
fact(N1, F1).

� Vigyázat!Az így előállókódmárnemfoglalkozik aszámokPeano-alakra
hozásával:

| ?- fact(N, 120). --> no
| ?- {F=120}, fact(N, F). --> F = 120, N = 5 ? ; no

Megjegyzés:a faktoriális példábannincsmérhető gyorsulás

CLP(MiniNat) javított változatai

A nulla szorzatproblémája

| ?- {X*X=0}.
X = 0 ? ; X = 0 ? ; no

A probléma1. javítása

% X*Y=0, ahol X és Y Peano számok.
szorzatuk_nulla(X, Y) :-

( X = 0
; X \== Y, Y = 0
).

| ?- {X*X=0}.
X = 0 ? ; no

| ?- {X*Y=0}, X=Y.
X = 0, Y = 0 ? ;
X = 0, Y = 0 ? ; no

A probléma2. javítása

% X*Y=0, ahol X és Y Peano számok.
szorzatuk_nulla(X, Y) :-

( X = 0
; dif(X, 0), Y = 0
).

| ?- {X*Y=0}, X=Y.
X = 0, Y = 0 ? ; no

CLP(MiniNat) javított változatai (folyt)

Az erőforrás probléma
� A fact(N, 11) hívásamásodikklózzalillesztve a {11=N*F1}

feltételrevezet̋odik vissza.Ezkétmegoldástgenerál(N=1,F1=11 , ill.
N=11,F1=1 . Ezekreabehelyettesítésekrefelébreda rekurzívfact hívás
előszöra fact(0,11) majda fact(10,1) paraméterekkel.
� A fact/2 másodikklózaezutóbbitmohónértékeli ki: kiszámolná10! -t,

éscsakezutánegyesítené1-gyel. Azonbana10! kiszámolásához(Peano
számként)kevésamemória:-( .
� A problémajavítása:aszorzat-feltételttegyüka rekurzívfact/2 híváselé.

:- block fact(-,-).
fact(N, F) :- {N = 0, F = 1}.
fact(N, F) :-

{N >= 1, N1 = N-1, F = N*F1},
fact(N1, F1).

| ?- fact(N, 24). --------> N = 4 ? ; no

� Azonbanazalábbicél futásamégígy is kivárhatatlan. . .

| ?- fact(N, 720). --------> N = 6 ? ;

Megjegyzések
� Egykorlát-programbanminél kés̋obbcélszer̋u választásipontotcsinálni.
� Ideálisancsakazösszeskorlát felvételeutánkezdjükmeg akeresést.
� Megoldás:egy külön keresésifázis(azún.címkézés,labeling):

program :-
korlátok_felvétele(...), labeling([V1, ..., VN]).

� CLP(MiniNat)-banazismertetetteszközökkel eznehezenmegoldható,de
� CLP(MiniB) esetén(lásd1. kis házifeladat)könnyenkészíthet̋o ilyen

labeling/1 eljárás.

1. kis házi feladat: CLP(MiniB) megvalósítása

CLP(MiniB) jellemzése
� Tartomány: logikai értékek (1 és0, igazéshamis)
� Függvények (egybenkorlát-relációk):

˜ P P hamis(negáció).
P * Q P ésQmindegyike igaz(konjunkció).
P + Q P ésQ legalábbegyike igaz(diszjunkció).
P # Q P ésQpontosanegyike igaz(kizáróvagy).
P =\= Q Ugyanazmint P # Q.
P =:= Q Ugyanazmint ~(P # Q) .

A megvalósítandóeljárások
� sat( Kif) , aholKif változókból,a0, 1 konstansokbóla fenti müveletekkel

felépítettlogikai kifejezés.Jelentése:A Kif logikai kifejezésigaz.A sat/1
eljárásnehozzonlétreválasztásipontot! A benneszerepl̋o változók
behelyettesítéseeseténminélelőbbébredjenfel, ésvégezzeel amegfelelő
következtetéseket (lásdapéldákatalább)!
� count( Es, N) , aholEsegy (változó-)lista,N adotttermészetesszám.

Jelentése:Az Es listábanpontosanN olyanelemvan,amelynekértéke1.
� labeling( Változók) . BehelyettesítiaVáltozókat0, 1 értekekre.

Visszalépéseseténfelsoroljaazösszeslehetségesértéket.

Futási példák

| ?- sat(A*B =:= (~A)+B).
---> <...felfüggesztett célok...> ? ; no

| ?- sat(A*B =:= (~A)+B), labeling([A,B]).
---> A = 1, B = 0 ? ; A = 1, B = 1 ? ; no

| ?- sat((A+B)*C=\=A*C+B), sat(A*B).
---> A = 1, B = 1, C = 0 ? ; no

| ?- count([A,A,B], 2). ---> <...felfüggesztett célok...> ? ; no
| ?- count([A,A,B], 2), labeling([A]).

---> A = 1, B = 0 ? ; no
| ?- count([A,A,B,B], 3), labeling([A,B]).

---> no
| ?- sat(~A =:= A). ---> no

1. kis házi feladat: egykis segítség

:- op(100, fx, ~).

~(A, B) :-
when( (nonvar(A); nonvar(B); ?=(A,B)),

not(A,B)
).

not(A, NA) :-
( nonvar(A) -> NA is 1-A
; nonvar(NA) -> A is 1-NA
; A == NA -> fail
).

| ?- trace, ~(A, A).
1 1 Call: ~(A,A) ?
2 2 Call: when((nonvar(A);nonvar(A) ;?=( A,A )),n ot(A ,A) ) ?
3 3 Call: not(A,A) ?
4 4 Call: nonvar(A) ?
4 4 Fail: nonvar(A) ?
5 4 Call: nonvar(A) ?
5 4 Fail: nonvar(A) ?
6 4 Call: A==A ?
6 4 Exit: A==A ?
3 3 Fail: not(A,A) ?
2 2 Fail: when((nonvar(A);nonvar(A) ;?=( A,A )),n ot(A ,A) ) ?
1 1 Fail: ~(A,A) ?

no
| ?- sat(A*A=:=B).

B = A ? ; no
| ?- sat(A#A=:=B).

B = 0 ? ; no
| ?- sat(A+B=:=C), A=B.

B = A, C = A ? ; no

18



CLP rendszerek a nagyvilágban

Néhányimplementáció

� clp(R)— azels̋o CLP(X) rendszer(MonashUniv, Australia,IBM Yorktown
HeightsésCMU)
� CHIP— FD, Q ésB (ECRC,München,Cosytec,Franciao.);CHARME

(Bull); DecisionPower (ICL)
� PrologIII, PrologIV (PrologIA,Marseille),Q (nem-lineárisis), B, FD,

listák, intervallumok
� ILOG solver (ILOG, Franciao.)— C++ könyvtár: R (nem-lineárisis), FD,

halmazok
� SICStusProlog(SICS,Svédo.)— R/Q,FD, B, CHR
� GNU Prolog(INRIA, Franciao.)— FD (C-refordít)
� Oz(DFKI, Németo.)— korlátalapúelosztottfunkcionálisnyelv.

Kommerciális rendszerek (a fentiek között)

� ILOG, CHIP, PrologIII–IV, SICStus
� aszakmaóriása:ILOG

– szakterület:CLP + vizualizációseszközök+ szabályalapúeszközök

– felvásároltaazegyik vezet̋o operációkutatásicéget,aCPLEX-et

– 400munkatárs7 országban

– 55M USDévesbevétel

– NASDAQ-onjegyzett

Mir ehasználják a CLP rendszereket

Ipari erőforrás optimalizálás

� termék-ésgépkonfiguráció
� gyártásütemezés
� emberierőforrásokütemezése
� logisztikaitervezés

Közlekedés,szállítás

� repül̋otériallokációsfeladatok(beszállókapu,poggyász-szalagstb.)
� repül̋o-személyzetjáratokhozrendelése
� menetrendkészítés
� forgalomtervezés

Távközlés,elektronika

� GSM átjátszókfrekvencia-kiosztása
� lokálismobiltelefon-hálózattervezése
� áramkörtervezésésverifikálás

Egyéb

� szabászatialkalmazások
� grafikusmegjelenítésmegtervezése
� multimédiaszinkronizáció
� légifelvételekelemzése

20



A CLP(
�

) séma

Egy adott CLP(
�

) meghatározásakor megkell adni

� akorlát-következtetéstartományát,
� akorlátokszintaxisátésjelentését(függvények,relációk),
� akorlát-megoldóalgoritmust.

A korlátok osztályozása

� egyszer̋u korlátok— akorlát-megoldóazonnaltudjakezelniőket;
� összetettkorlátok— felfüggesztve,démonkéntvárnakarra,hogya

korlát-megoldónaksegíthessenek.

A CLP(
�

) korlát-megoldók közösvonása:a korlát tár

� A korlát tárkonzisztenskorlátokhalmaza(konjunkciója).
� A korlát tárelemeiegyszer̋u korlátok.
� A közönségesPrologvégrehajtássoránakurrenscélsorozatmelletta

CLP(
�

) rendszernyilvántartjaakorlát tárállapotát:

– amikor avégrehajtásegy egyszer̋u korláthozér, akkor aztamegoldó
megpróbáljahozzávennia tárhoz;

– haazúj korláthozzávételével a tár konzisztensmarad,akkor eza
redukcióslépéssikeresésa tár kibővül azúj korláttal;

– haazúj korláthozzávételével a tár inkonzisztensséválna,akkor (nem
kerül bea tárbaés)meghiúsulást,azazvisszalépéstokoz;

– visszalépéseseténakorlát tár is visszaállakorábbiállapotába.
� aösszetettkorlátokdémonként(ágensként)várakoznakarra,hogy:

a. egyszer̋u korláttáváljanak

b. a tárategy egyszer̋u következményükkel bővíthessék(azún.erősítés)

A SICStusclp(Q,R) könyvtárak

A clpq /clpr könyvtárak

� Tartomány:

– clpr : lebegőpontosszámok

– clpq : racionálisszámok
� Függvények:

+ - * / min max pow exp (kétargumentumúak,pow � exp ),

+ - abs sin cos tan (egyargumentumúak).
� Korlát-relációk:

= =:= < > =< >= =\= (= � =:= )
� Primitív korlátok(korlát tárelemei):

lineáriskifejezéseket tartalmazórelációk
� Korlát-megoldóalgoritmus:

lineárisprogramozásimódszerek:Gausselimináció,szimplex módszer

A könyvtár betöltése:

� use_module(library(clpq )) , vagy
� use_module(library(clpr ))

A fő beépítetteljárás

� { Korlát } , aholKorlát változókbólés(egészvagylebegőpontos)
számokbóla fenti műveletekkel felépítettreláció,vagyilyen relációknaka
vessz̋o (, ) operátorralképzettkonjunkciója.

Példafutása SICStusclpq könyvtárával

Példafutás

| ?- use_module(library(clpq )).
{loading .../library/clpq.ql...}
...

| ?- {X=Y+4, Y=Z-1, Z=2*X-9}.
X = 6, Y = 2, Z = 3 ? % lineáris egyenlet

| ?- {X+Y+9<4*Z, 2*X=Y+2, 2*X+4*Z=36}.
% lineáris egyenl őtlenség

{X<29/5}, {Y= -2+2*X}, {Z=9-1/2*X} ?
% az eredmény: a tár állapota

| ?- {(Y+X)*(X+Y)/X = Y*Y/X+100}.
{X=100-2*Y} ? % lineárissá egyszer˝ usíthet ő

| ?- {(Y+X)*(X+Y) = Y*Y+100*X}.
% így már nem lineáris

clpq:{2*(X*Y)-100*X+X^2=0 } ?
% a clpq modul-prefix jelzi, hogy
% felfüggesztett összetett
% hívásról van szó

| ?- {exp(X+Y+1,2) = 3*X*X+Y*Y}.
% nem lineáris...

clpq:{1+2*X+2*(Y*X)-2*X^2 +2* Y=0} ?

| ?- {exp(X+Y+1,2) = 3*X*X+Y*Y}, X=Y.
X = -1/4, Y = -1/4 ? % így már igen...

| ?- {2 = exp(8, X)}. % nem-lineárisak is
% megoldhatók

X = 1/3 ?

Összetettkorlátok kezeléseCLP(Q)-ban

Példavárakozóágensre

| ?- {X =< Y}, {X*(Y+1) > X*X+Z},
( Z = X*(Y-X), {Y < 0}
; Y = X
).

Y = X, {X-Z>0} ? ; no

A végrehajtáslépései

| ?- {X =< Y}, {X*(Y+1) > X*X+Z}.
{X-Y=<0}, clpq:{Z-X-Y*X+X^2<0} ?

| ?- {X =< Y}, {X*(Y+1) > X*X+Z}, Z = X*(Y-X).
Z = X*(Y-X), {X-Y=<0}, {X>0} ?

| ?- {X =< Y}, {X*(Y+1) > X*X+Z}, Z = X*(Y-X), {Y < 0}.
no

| ?- {X =< Y}, {X*(Y+1) > X*X+Z}, Y = X.
Y = X, {X-Z>0} ?

Példaegy lehetségeserősítésilépésre
� A tár tartalma:X > 3.
� A végrehajtandóösszetettkorlát: Y > X*X .
� A korlátotaCLPmegoldónemtudjafelvennia tárba,deegy

következményét, pl. azY > 9 korlátotfelvehetné!
� Az erősítésutánazeredetiösszetettkorlát továbbrais démonkéntkell

lebegjen!
� Fontosmegjegyzés:aCLP(Q/R)rendszernemhajtjavégrea fenti

következtetést,ésáltalánosansemmiféleerősítéstnemvégez.

Egy összetettebb példa: hiteltörlesztés

% Hiteltörlesztés számítása: P összeg˝u hitelt
% Time hónapon át évi IntRate kamat mellett havi MP
% részletekben törlesztve Bal a maradványösszeg.
mortgage(P, Time, IntRate, Bal, MP):-

{Time > 0, Time =< 1,
Bal = P*(1+Time*IntRate/1200)-T ime* MP}.

mortgage(P, Time, IntRate, Bal, MP):-
{Time > 1},
mortgage(P*(1+IntRate/1 200) -MP,

Time-1, IntRate, Bal, MP).

| ?- mortgage(100000,180,12, 0,MP).
% 100000 Ft hitelt 180
% hónap alatt törleszt 12%-os
% kamatra, mi a havi részlet?

MP = 1200.1681 ?

| ?- mortgage(P,180,12,0,120 0).
% ugyanez visszafelé

P = 99985.9968 ?

| ?- mortgage(100000,Time,12 ,0,1 300) .
% 1300 Ft a törleszt őrészlet,
% mi a törlesztési id ő?

Time = 147.3645 ?

| ?- mortgage(P,180,12,Bal,M P).

{MP=0.0120*P-0.0020*Bal} ?

| ?- mortgage(P,180,12,Bal,M P), ordering([P,Bal,MP]).

{P=0.1668*Bal+83.3217*MP} ?

További könyvtári eljárások

entailed(Korlát) — Korlát levezethet̋o a jelenlegi tárból.

inf(Kif, Inf) ill. sup(Kif, Sup) — kiszámoljaKif infímumátill.
szuprémumát,ésegyesítiInf -fel ill. Sup-pal. Példa:

| ?- { 2*X+Y =< 16, X+2*Y =< 11, X+3*Y =< 15,
Z = 30*X+50*Y

}, sup(Z, Sup).

Sup = 310, {....}

minimize(Kif) ill. maximize(Kif) — kiszámoljaKif infimumátill.
szuprémumát,ésegyenl̋ovétesziKif -fel. Példa:

| ?- { 2*X+Y =< 16, X+2*Y =< 11, X+3*Y =< 15,
Z = 30*X+50*Y

}, maximize(Z).

X = 7, Y = 2, Z = 310

bb_inf(Egészek, Kif, Inf) — kiszámoljaKif infimumát,azzala
továbbifeltétellel,hogyazEgészeklistábanlevő mindenváltozóegész(ún.
„Mix edIntegerOptimisationProblem”).

| ?- {X >= 0.5, Y >= 0.5}, inf(X+Y, I).

I = 1, {Y>=1/2}, {X>=1/2} ?

| ?- {X >= 0.5, Y >= 0.5}, bb_inf([X,Y], X+Y, I).

I = 2, {X>=1/2}, {Y>=1/2} ?

ordering(V1 < V2) — A V1 változóelőbbszerepeljenaz
eredmény-korlátbanmint aV2 változó.

ordering([V1,V2,...]) — V1, ... ebbenasorrendbenszerepeljenaz
eredmény-korlátban.

Széls̋oérték-számításgrafikus illusztrálása

2x+y=<16

x+3y=<15x+2y
=<11

310=30x+50y

| ?- { 2*X+Y =< 16, X+2*Y =< 11, X+3*Y =< 15,
Z = 30*X+50*Y

}, sup(Z, Sup).

Sup = 310, {Z=30*X+50*Y},
{X+1/2*Y=<8}, {X+3*Y=<15}, {X+2*Y=<11}

További részletek

Projekció

% Az (X,Y) pont az (1,2) (1,4) (2,4) pontok
% által kifeszített háromszögben van.
hszogben(X,Y) :-

{ X=1*L1+1*L2+2*L3,
Y=2*L1+4*L2+4*L3,
L1+L2+L3=1, L1>=0, L2>=0, L3>=0 }.

| ?- hszogben(X,Y).
{Y=<4}, {X>=1}, {X-1/2*Y=<0} ?

| ?- hszogben(_, Y).
{Y=<4}, {Y>=2} ?

| ?- hszogben(X, _).
{X>=1}, {X=<2} ?

Bels̋o ábrázolás
clpr — lebegőpontosszám;clpq —- rat( Számláló, Nevez̋o) , ahol
SzámlálóésNevez̋o relatívprímek.Példáulclpq -ban:

| ?- {X=0.5}, X=0.5.
no
| ?- {X=0.5}, X=1/2.
no
| ?- {X=0.5}, X=rat(2,4).
no
| ?- {X=0.5}, X=rat(1,2).
X = 1/2 ?
| ?- {X=5}, X=5.
no
| ?- {X=5}, X=rat(5,1).
X = 5 ?

Egy nagyobb CLP(Q) feladat: Tökéletestéglalapok

A feladat

� egy olyantéglalapkeresése
� amelykirakhatópáronkéntkülönböz̋o oldalúnégyzetekb̋ol

Egy megoldás
(a legkevesebb,9 darabnégyzetfelhasználásával)

18
15

14

4
78

9
10

32

33

1

Tökéletestéglalapok — CLP(Q) megoldás

% Colmerauer A.: An Introduction to Prolog III,
% Communications of the ACM, 33(7), 69-90, 1990.

% Rectangle 1 x Width is covered by distinct
% squares with sizes Ss.
filled_rectangle(Width, Ss) :-

{ Width >= 1 }, distinct_squares(Ss),
filled_hole([-1,Width,1], _, Ss, []).

% distinct_squares(Ss): All elements of Ss are distinct.
distinct_squares([]).
distinct_squares([S|Ss]) :-

{ S > 0 }, outof(Ss, S), distinct_squares(Ss).

outof([], _).
outof([S|Ss], S0) :- { S =\= S0 }, outof(Ss, S0).

% filled_hole(L0, L, Ss0, Ss): Hole in line L0
% filled with squares Ss0-Ss (diff list) gives line L.
% Def: h(L): sum of lengths of vertical segments in L.
% Pre: All elements of L0 except the first >= 0.
% Post: All elems in L >=0, h(L0) = h(L).
filled_hole(L, L, Ss, Ss) :-

L = [V|_], {V >= 0}.
filled_hole([V|HL], L, [S|Ss0], Ss) :-

{ V < 0 }, placed_square(S, HL, L1),
filled_hole(L1, L2, Ss0, Ss1), { V1=V+S },
filled_hole([V1,S|L2], L, Ss1, Ss).

% placed_square(S, HL, L): placing a square size S on
% horizontal line HL gives (vertical) line L.
% Pre: all elems in HL >=0
% Post: all in L except first >=0, h(L) = h(HL)-S.
placed_square(S, [H,V,H1|L], L1) :-

{ S > H, V=0, H2=H+H1 },
placed_square(S, [H2|L], L1).

placed_square(S, [S,V|L], [X|L]) :- { X=V-S }.
placed_square(S, [H|L], [X,Y|L]) :-

{ S < H, X= -S, Y=H-S }.

Tökéletestéglalapok: példafutás

% 600 MHz Pentium III
| ?- length(Ss, N), N > 1, statistics(runtime, _),

filled_rectangle(Width, Ss),
statistics(runtime, [_,MSec]).

N = 9, MSec = 8010, Width = 33/32,
Ss = [15/32,9/16,1/4,7/32,1/ 8,7/ 16,1 /32, 5/1 6,9/ 32] ? ;

N = 9, MSec = 1010, Width = 69/61,
Ss = [33/61,36/61,28/61,5/61 ,2/6 1,9/ 61,2 5/6 1,7/ 61,1 6/6 1] ? ;

N = 9, MSec = 10930, Width = 33/32,
Ss = [9/16,15/32,7/32,1/4,7/ 16,1 /8,5 /16, 1/3 2,9/ 32] ?

Az outof híváskihagyásával végzettfuttatás

Kommentkéntközöljükazadottágongeneráltkorlátokat,a redundánsak
elhagyásával.

| ?- filled_rectangle(W, [S1,S2,S3], [eqsq]).

S1 = 1/2, S2 = 1, S3 = 1/2, W = 3/2 ? ; % 3 3 2 2 2 2
% 3 3 2 2 2 2

% {W=S1+S2}, {S2=<1}, {S1=S3}, % 1 1 2 2 2 2
% {S2>=S1+S3}, {S1+S3>=1}. % 1 1 2 2 2 2

S1 = 1, S2 = 1/2, S3 = 1/2, W = 3/2 ? ; % 1 1 1 1 3 3
% 1 1 1 1 3 3

% {W=S1+S2}, {S2=S3}, {S2+S3=<1}, % 1 1 1 1 2 2
% {S2+S3>=S1}, {S1>=1}. % 1 1 1 1 2 2

S1 = 1, S2 = 1, S3 = 1, W = 3 ? ; no

% {W=S1+S2+S3}, {S3=<1}, {S3>=S2}, % 1 1 2 2 3 3
% {S2>=S1}, {S1>=1}. % 1 1 2 2 3 3

Tökéletestéglalapok: választásipontok

Függőleges

V1=<V2 V1>V2

Függ.vál.

V1
V2

Vízszintes

S

H

H1

V

Vízsz. vál.
V=0,
S>H

S=H

S<H

Tökéletestéglalapok: a keresésitér szerkezete

?

?

?

?>0

zsákutca

S4>=S3

S2>=S1

?=0

S2<S1

S4<S3

?

?

Függ. vál.

Vízsz. vál.

Függ. vál.

S1

S1
S2

S1
S2

S3
S4

S2

S6

S5
S4S3

S9 S8

S7

S1

W

1

A korlát logikai programozáselmélete

Egy CLP rendszer

���	��
��
���
���
��� : egy tartomány (domain),pl. egészek(N), valósak(R), racionálisak(Q),

Booleértékek (B), listák, füzérek(stringek)(+ aProlog-fastruktúrák
(Herbrand— H) tartománya)
��� : � -bendefiniáltfüggvényjeleknekegy halmaza,pl. � , � , � , � , �
��� : � -bendefiniáltrelációjeleknek(korlátoknak)egy halmazapl.

�
, �� , � , �

��� : egy korlát-megoldóalgoritmus ����
���
�� � -re,azaza � tartománybanaz� !"� halmazbelijelekb̋ol felépítettkorlátokra

CLP szintaxisésdeklaratív szemantika

program

� klózokhalmaza.

klóz

� szintaxis:P :- G# , . . . , G$ , aholmindegyik G% vagyeljáráshívás,vagy
korlát.
� deklaratívolvasat:P igaz,haG# , . . . , G$ mind igaz.

kérdés

� szintaxis:?- G# , . . . , G$
� válaszegy Qkérdésre:korlátoknakegy olyankonjunkciója,amelyb̋ol a

kérdéskövetkezik.

34



CLP procedurálisszemantika

Végrehajtásiállapot

��� G, & �
� G— cél/korlát sorozat
� & — korlát-tár:azeddigfelhalmozottegyszer̋u korlátokkonjunkciója

(kezdetbenüres)

Szükségesmegkülönböztetés

� egyszer̋u korlát (c ): amitakorlát-tárközvetlenülbefogad( � !�� -től függ)
� összetettkorlát (C): a tárnemtudjabefogadni,dehathata tárra

Klózok procedurálisolvasata

� P :- G# , . . . , G$ jelentése:P megoldásáhozmegoldandóG# , . . . , G$ .

Végrehajtási invariánsok

� & konzisztens
� G �'&�( Q(Qakezd̋o kérdés)

Végrehajtásvége

��� G) , &*) � , aholG) -renemalkalmazhatóegyetlenkövetkeztetésilépéssem.

A végrehajtáseredménye

� Az &*) korlát-tár, vagyannakakérdésbenszerepl̋o változókravaló „vetítése”
(a többi változóegzisztenciáliskvantálásával).
� A G) fennmaradó(összetett)korlátok.

A CLP következtetésfolyamata

Következtetésilépések

� rezolúció:� P & G, & �,+ � G# & . . . & G$ & G, P = P-.�'& � ,
feltéve,hogyaprogrambanvanegy P- :- G# , . . . , G$ klóz
� korlát-megoldás:� c & G, & �,+ � G, &�� c �
� korlát-er̋osítés:� C& G, & �,+ � C- & G, &�� c �

ha & -ből következik,hogyCekvivalens(C-/� c )-vel. (C- = C is lehet.)

Haa tár inkonzisztenséválna,visszalépéstörténik.

Példaerősítésre

��� X > Y*Y & . . . , Y > 3 �0+ � X > Y*Y & . . . , Y > 3 � X > 9 �
hiszenX > Y*Y � Y > 3 + X > 9

� clp(R)-bennincsilyen, declp(FD)-benvan!

Követelmények a korlát megoldóalgoritmussal szemben

� teljesség(egyszer̋u korlátokkonjunkciójárólmindig döntseel, hogy
konzisztens-e),
� inkrementalitás(az & tár konzisztenciájátnebizonyítsaújra),
� avisszalépéstámogatása,
� hatékonyság.

A clpb könyvtár

� Tartomány: logikai értékek (1 és0, igazéshamis)
� Függvények (egybenkorlát-relációk):

˜ P P hamis(negáció).
P * Q P ésQmindegyike igaz(konjunkció).
P + Q P ésQ legalábbegyike igaz(diszjunkció).
P # Q P ésQpontosanegyike igaz(kizáróvagy).
X ˆ P LétezikolyanX, hogyP igaz

(azazP[X/0]+P[X/1] igaz).
P =\= Q Ugyanazmint P # Q.
P =:= Q Ugyanazmint ~(P # Q) .
P =< Q Ugyanazmint ~P + Q.
P >= Q Ugyanazmint P + ~Q.
P < Q Ugyanazmint ~P * Q.
P > Q Ugyanazmint P * ~Q.
card(Is, Es) Az Es listábanszerepl̋o igaz érték̋u kifejezések

számaelemeazIs által jelölt halmaznak(Is egé-
szekésTol-Ig szakaszoklistája).

� Primitív korlátok (korlát tárelemei):tetsz̋olegeskorlát (Boole-egyesít̋ok
formájában).
� Korlát-megoldóalgoritmus: Boole-egyesítés.

A library(clpb) könyvtár eljárásai

� sat( Kifejezés) , aholKifejezésváltozókból,a0, 1 konstansokbólés
atomokból(ún. szimbolikuskonstansok)a fenti müveletekkel felépített
logikai kifejezés.HozzávesziKifejezést a korlát-tárhoz.
� taut( Kif, Ért) . Megvizsgálja,hogyKif levezethet̋o-ea tárból,ekkor

Ért=1; vagynegáltjalevezethet̋o-e,ekkor Ért=0. Egyébkéntmeghiúsul.
� labeling( Változók) . BehelyettesítiaVáltozókat0, 1 értekekre(úgy,

hogya tár teljesüljön).Visszalépéskor felsoroljaazösszeslehetségesértéket.

Egyszer̋u példák

| ?- sat(X + Y). sat(X=\=_A*Y#Y) ?

| ?- sat(x + Y). sat(Y=\=_A*x#x) ?

| ?- taut(_A ^ (X=\=_A*Y#Y) =:= X+Y, T).
T = 1 ?

| ?- sat(A # B =:= 0). B = A ?

| ?- sat(A # B =:= C), A = B. B = A, C = 0 ?

| ?- taut(A =< C, T). no

| ?- sat(A =< B), sat(B =< C), taut(A =< C, T).
T = 1,
sat(A=:=_A*_B*C),
sat(B=:=_B*C) ?

Megjegyzések

� A tármegjelenítése:sat(V =:= Kif) ill. sat(V = 1 = Kif) aholKif
egy „polinom”, azazkonjunkciókbólkizáróvagy(#) művelettelképzett
kifejezés.
� Az atommaljelölt szimbolikuskonstansoknembehelyettesíthetőek,

(legkívül) univerzálisankvantifikált változóknaktekinthet̋ok.

| ?- sat(~x+ ~y=:= ~(x*y)). % 2 xy 354 x ��4 y
� 463 x � y 7�7

yes
| ?- sat(~X+ ~Y=:= ~(X*Y)). % 8:9 XY354 X �;4 Y

� 463 X � Y7�7
true ? ; no

| ?- sat(x=<y). % 2 xy 3 x ( y 7
no

| ?- sat(X=<y). % 2 y 8:9 X 3 X ( y 7
sat(X=:=_A*y) ? ; no

Példa: 1-bitesösszeadó

| ?- [user].
| adder(X, Y, Sum, Cin, Cout) :-

sat(Sum =:= card([1,3],[X,Y,Cin])),
sat(Cout =:= card([2-3],[X,Y,Cin])).

| {user consulted, 40 msec 576 bytes}

yes
| ?- adder(x, y, Sum, cin, Cout).

sat(Sum=:=cin#x#y),
sat(Cout=:=x*cin#x*y#y*ci n) ?

yes
| ?- adder(x, y, Sum, 0, Cout).

sat(Sum=:=x#y),
sat(Cout=:=x*y) ?

yes
| ?- adder(X, Y, 0, Cin, 1), labeling([X,Y,Cin]).

Cin = 0, X = 1, Y = 1 ? ;

Cin = 1, X = 0, Y = 1 ? ;

Cin = 1, X = 1, Y = 0 ? ;

no

Boole-egyesítés

A feladat:
� Adott g ésh logikai kifejezések.
� Keressükag = h egyenletetmegoldólegáltalánosabbegyesít̋ot (mgu).
� Példa:mgu(X+Y, 1) lehetX = W * Y # Y # 1 (új változó,pl. W,

bejöhet).
� Egyszer̋usítés:A g = h egyenlethelyettesíthet̋o azf = 0 egyenlettel,

aholf = g # h.
� Az egyesítéssoránmindenlépésbenegy < = 0 formulabeliváltozót

szeretnénkkifejezni.

Az X változókifejezése
� Legyen <>=?3@A7 az < -ből azX=1, <>=?3CB.7 azX=0 behelyettesítésselkapott

kifejezés.
� < = 0 kielégíthet̋oségénekszükségesfeltétele < = 3@A7 * < = 3CB.7 = 0

kielégíthet̋osége.
� Fejezzükki X-et < = 3CB.7 -val és < = 3�@>7 -gyelúgy, hogy < = 0 legyen!

< = 3CB.7 < = 3@A7 X
0 0 bármi(W)
0 1 0
1 0 1
1 1 érdektelen

KeressükX-etX = A*˜W # B*Walakban!
� Határozzukmeg A-t ésB-t < = 3DBE7 és < = 3@A7 függvényeként!

< = 3CB.7 < = 3@A7 X A B
0 0 W 0 1
0 1 0 0 0
1 0 1 1 1

Az A = <>=F3CB.7 ésB = ˜ <A=G3�@>7 megfeleltetéstűnik a legegyszer̋ubbnek.

Boole-egyesítés(folyt.)

Az egyesítésialgoritmus az < = 0 egyenlőségre

� Ha < -bennincsváltozó,akkor azonosnakkell lennie0-val (különbennem
egyesíthet̋o).
� Helyettesítsünk:X = ˜W* <A=H3CB.7 # W*˜ <A=G3@A7 (Boole-egyesít̋o)
� Folytassukazegyesítéstaz <A=F3@A7 * <A=G3CB.7 = 0 egyenl̋oségre.

Példák

� mgu(X+Y, 0) �I( X = 0, Y = 0;
� mgu(X+Y, 1) = mgu(̃ (X+Y) , 0) �J( X = W * Y # Y # 1;
� mgu(X*Y , ˜(X*Z) ) = mgu((X*Y)#(X*Z)#1 , 0) �J( X = 1, Y = ˜Z .

Bels̋o ábrázolás: BDD (Boolean/BinaryDecisionDiagrams)

(Szaggatottvonal:0 érték,folytonosvonal: 1 érték)

(X+Y) # 1 X*Y # X*Z # 1

Y

1

0

X

1 0

Z

Y

X

Z

41



Példa: Hibakeresésáramkörben

X

Sum

Cout

Cin

Y

U1

U3
U2

fault([F1,F2,F3,F4,F5], [X,Y,Cin], [Sum,Cout]) :-
sat(

card([0-1],[F1,F2,F3,F4 ,F5] ) *
(F1 + (U1 =:= X * Cin)) *
(F2 + (U2 =:= Y * U3)) *
(F3 + (Cout =:= U1 + U2)) *
(F4 + (U3 =:= X # Cin)) *
(F5 + (Sum =:= Y # U3))

).

| ?- fault(L, [1,1,0], [1,0]).
L = [0,0,0,1,0] ? ; no

| ?- fault(L, [1,0,1], [0,0]).
L = [_A,0,_B,0,0],
sat(_A=\=_B) ? ; no

| ?- fault(L, [1,0,1], [0,0]), labeling(L).
L = [1,0,0,0,0] ? ;
L = [0,0,1,0,0] ? ; no

| ?- fault([0,0,0,0,0], [x,y,cin], [Sum,Cout]).
sat(Cout=:=x*cin#x*y#y*cin ),
sat(Sum=:=cin#x#y) ? ; no

Példa: Tranzisztorosáramkör verifikálása

A

B

Out
X

1

0

n(D, G, S) :- % Gate => Drain = Source
sat( G*D =:= G*S).

p(D, G, S) :- % ~ Gate => Drain = Source
sat( ~G*D =:= ~G*S).

xor(A, B, Out) :-
p(1, A, X),
n(0, A, X),
p(B, A, Out),
n(B, X, Out),
p(A, B, Out),
n(X, B, Out).

| ?- n(D, 1, S). S = D ?

| ?- n(D, 0, S). true ?

| ?- p(D, 0, S). S = D ?

| ?- p(D, 1, S). true ?

| ?- xor(a, b, X). sat(X=:=a#b) ?

Minesweeperclpb -ben

:- use_module([library(clpb),libra ry(l ists) ]).

mine(Rows, Cols, Mines, Bd) :-
length(Bd, Rows), all_length(Bd, Cols),
append_lists(Bd, All),
sat(card([Mines], All)), play_mine(Bd, []).

all_length([], _).
all_length([L|Ls], Len) :-

length(L, Len), all_length(Ls, Len).

append_lists([], []).
append_lists([L|Ls], Es) :-

append_lists(Ls, Es0), append(L, Es0, Es).

play_mine(Bd, Asked) :-
select_field(Bd, Asked, R, C, E), !,
format(’Row ~w, col ~w (m for mine)? ’, [R,C]),
read(Ans), process_ans(Ans, E, R, C, Bd),
play_mine(Bd, [R-C|Asked]).

play_mine(_Bd, _Asked).

select_field(Bd, Asked, R, C, E) :-
nth(R, Bd, L), nth(C, L, E),
non_member(R-C, Asked), taut(E, 0), !.

select_field(Bd, _Asked, R, C, E) :-
nth(R, Bd, L), nth(C, L, E),
non_member(R-C, Asked), \+ taut(E,1), !.

process_ans(m, 1, _, _, _) :-
format(’Mine!~n’, []), !, fail.

process_ans(Ans, 0, R, C, Bd) :-
integer(Ans), neighbs(n(R, C, Bd), Ns),
sat(card([Ans], Ns)).

neighbs(RCB, N7) :-
neighbour(-1,-1, RCB, [], N0),
neighbour(-1, 0, RCB, N0, N1),
neighbour(-1, 1, RCB, N1, N2),
neighbour( 0,-1, RCB, N2, N3),
neighbour( 0, 1, RCB, N3, N4),
neighbour( 1,-1, RCB, N4, N5),
neighbour( 1, 0, RCB, N5, N6),
neighbour( 1, 1, RCB, N6, N7).

neighbour(ROf, COf, n(R0, C0, Bd), Nbs, [E|Nbs]) :-
R is R0+ROf, C is C0+COf,
nth(R, Bd, Row), nth(C, Row, E), !.

neighbour(_, _, _, Nbs, Nbs).

